Paper 6474-2016

A macro for generating a ternary graph for a ceramics industry process

Igor Ferreira do Nascimento, IFPI; Roberto Arruda Lima Soares, IFPI;
Zacarias Linhares, IFPI

ABSTRACT

This work presents a macro for generating a ternary graph that can be used to solve a problem in the ceramics industry. The ceramics industry uses mixtures of clay types to generate a product with good properties that can be assessed according to breakdown pressure after incineration, the porosity, and water absorption. Beyond these properties, the industry is concerned with managing geological reserves of each type of clay. Thus, it is important to seek alternative compositions that present properties similar to the default mixture. This can be done by analyzing the surface response of these properties according to the clay composition, which is easily done in a ternary graph. SAS® documentation does not describe how to adjust an analysis grid in nonrectangular forms on graphs. A macro Triaxial, however, can generate ternary graphical analysis by creating a special grid, using an annotated database, and making linear transformations of the original data. The GCONTOUR procedure is used to generate three-dimensional analysis.

1 - INTRODUCTION

The response surface analysis is a graphic resource for understanding mathematical statistical models that use more than one dependent variable. In industry, the manufacturing process has essentially two components: the raw material and the final product, which can be seen as independent and dependent variables, respectively. In this context, the response surface analysis may be used to study the characteristics of the product according to the variations of the mixtures used as raw material.

In the field of ceramic industry, the main raw material is clay, which can be found next to rivers often forming gullies on the banks. The clay can be found in different colors (white, yellow or red). In SAS® SYSTEM, there are many procedures to generate graphics with more than two dimensions, including PROC GCONTOUR and PROC G3D. Such procedure generates graphics capabilities that help analyze the data, and mark out decisions. However, for models in which the valid regions of analysis have strongly restriction, these procedures into SAS® do not provide resources to delimit this region study.

In this context, the potential of some graphics, such as ternary cannot be completely exploited. The aim of this paper is to present a Macro to generate graphs of the ternary type with visual aids to assist data modeling. This is possible by adjusting the original database through a specific linear transformation and using the PROC GCONTOUR to generate three-dimensional analysis. Furthermore, this work can help the analysts of industry to explore the model with ternary composition using creative graphics capabilities.

The paper is organized as follows. In Section 2, presents an industrial process, Section 3, the procedure to create a ternary graph is presented, Section 4, a macro is presented and Section 5, presents the macro results for ceramics industry.

2 - FIELD OF APPLICATION: ARGILS COMPOSITIONS

The industries which produce tiles and bricks use mixtures of several types of clays, with different colors (white, yellow or red) as raw material. This work, like Linhares (2015), considers three types of clay for the industrial process: A, B and C. In the manufacturing process been used a compounds clays of the three types, having usually, a larger amount of clay C. The default mixture is 25% A, 25% B and 50% Clay C, because believe that the clay Type C has better properties when compared with the others. These properties may be assessed by some product information, breakdown pressure after incineration, porosity and water absorption.

In addition to these properties the industry is concerned with managing geological reserves of clay, especially in relation to clay type C (highest percentage in the default composition). Thus, it is important to seek alternative compositions that present similar properties to the mixture default. This can be done by analyzing the response of these surface properties according to the clay composition.

The model for this ternary system the information is formed by composition only two clays, because third percentage is a linear combination of the others. That is, if the sum of three proportions is equal to 1, being "a" the proportion of clays type "A" , "b" being B, so, "1-a-b" is the proportion of C. This result is valid for any process that has a finite number of components.

Thus, it is possible to represent the system in a two-dimensional graph. It is important to note that there are non valid domain regions. The Figure 1 shows the two-dimensional representation of valid domain using clays A and B.



Figure 1. Grid of valid domain representation for the model. The percentage composition of the mixture is shown on the abscissa axis the clay type A and the axis of ordinates type B. So the composition of clay type C is equal to the remaining percentage.

3 - TERNARY GRAPH

The ternary graph can be used for instance with are three information that sum is equal a constant. Areas as chemistry, biology, mining, metallurgy industries have been wide use for represents mixtures of elements or amount of species in an ecosystem.

For do this representation is necessary to use a special grid, inserted in an equilateral triangle. Each side of the triangle is associated with information that is expressed with parallel strips. For create the triangle region need to perform linear transformations on the original region. The Figure 2 is the ternary representation of the parameter space shown in Figure 1.



Figure 2. The grid transformed for ternary model. Each side of triangle represents one variable that is expressed with parallel trips according as raise percentage in mixture.

A linear transformation T is a type of function that establishes a relation to domain and co domain between two vector spaces. There are several types of linear transformations: reflection, rotation, translation, scaling (enlargement and reduction) and shear. Ternary region is performing the **shearing and reduction transformations**, expressed in Figure 3 and Figure 4.



Figure 3. Shear transformation of x axes with angle θ.



Figure 4. Shear transformation of x axes with angle θ.

Let be P = (x, y) a belonging point to region 1 and P '= (x', y ') the corresponding point in the region 2. The value x' is calculated by the shear transformation with θ = 30 °.

$$(x' \quad y') = (x \quad y)\begin{bmatrix} 1 & 0 \\ \sin(\theta) & 1 \end{bmatrix}$$

The value of y' is obtained by reduction transformation with λ = cos (30 °), it is the height of the equilateral triangle with side equal to 1.

$$(0 \quad y') = (x \quad y)\begin{bmatrix} 1 & 0 \\ 0 & \cos(\theta) \end{bmatrix}$$

Thus, the point P '= (x', y ') is obtained by combining the two transformations:



Figure 5. Grid transformation.

Thus, in this equilateral triangle is possible perform the three-dimensional analysis using the contour plot, widely used in topographic representation of maps and design of experiment. In this chart type, the third dimension is represented by forms with distinct colors.

4 - THE MACRO

The SAS® system, there are several procedures that perform 3D visualization, but we'll use the PROC GCONTOUR to generate the contour analysis. In this procedure, the grid is the main limitation for graphical analysis models with constraints in the response area.

For use the graphics of PROC GCONTOUR the area plot must be quadrilateral area grid the graphic analysis is generate just if observations of database must occupy at least half of the quadrilateral grid. In case with unperformed process the procedure displays the following message in the log.

> *"ERROR: Less than half of the grid cells have data values. No contour plot will be produced.*
> *The points in data set are less than half of grid"*

For overcome, the macro generate a quadrilateral grid and assigns the smallest value of database for non valid observations and use a background color for this area. For this, is important that the number of information be a sufficient quantity to describe the ternary grid. We suggest using the PROCEDURE G3GRID to create a big grid dataset, with interpolated estimates. The information of grid must be percentage format and sum to 1 with three variables (x, y and w). Thus, apply on grid generated and obtain the approximate response surface and estimate the response of study variable. The call function macro is:

```
% Triaxial(tab=, /*the name of database*/
            Var=, /*the name of the variable to be analyzed*/
            x=, /*the name of the variable in first axis of ternary graph*/
            y=, /*the name of the variable to be used to generate the second
                   axis of ternary graph*/
            w=, /*the name of the variable to be used to generate the third
                   axis of ternary graph*/
            ctl=, /*The vector with the cut lines of contour analysis*/
            col=, /*The vector with the colors of cut lines of contour
                   analysis*/
            pxx=95, /*Label position of first variable in axis x in annotate
                   coordinate system 1, 0-100*/
            pxy=5, /*Label position of first variable in axis y in annotate
                   coordinate system 1, 0-100*/
            pyx=50, /*Label position of second variable in axis x in annotate
                   coordinate system 1, 0-100*/
            pyy=97, /*Label position of second variable in axis y in annotate
                   coordinate system 1, 0-100*/
            pwx=5, /*Label position of third variable in axis x in annotate
                   coordinate system 1, 0-100*/
            pwy=5, /*Label position of third variable in axis y in annotate
                   coordinate system 1, 0-100*/
            help=1);/*Dichotomy variable to generate additional graphs to help
                   understand the ternary graph. 0: not and 1 yes, 1 default*/
```

The Macro adjusts the region analysis, transforming the region 1 to region 2, like the Figure 5, create the special grid and contour analysis.

5 - ILLUSTRATION

It was used for the application of Linhares (2015) data, which conducted an experiment to study the properties of the mixture of three types of clay in the production of elements of the ceramic industry: bricks and tiles. It was built a mathematical model to study characteristics and properties of tiles produced by the compound formed by clay mixture. For this, the author uses eight different compositions, as shown in Table 1 andFigure 6.

Design	A	B	C
A0	100%	0%	0%
B0	0%	100%	0%
C0	0%	0%	100%
F0	25%	25%	50%
F4	50%	50%	0%
F5	50%	0%	50%
F6	0%	50%	50%
F7	33,3%	33,3%	33,3%

Table 1. Eight designs of experiment.



Figure 6. Ternary graph of design model.

For use the Macro Triaxial was used the masters dissertation Linhares (2015) to generate the ternary graph model estimated for the breakdown pressure burning (Z1). Each blend composition of clays produced samples of tiles that measured breakdown pressure. The author proposes a linear model with quadratic terms to generate the approximate response surface of variable.

$$f(\Delta_A, \Delta_B, \Delta_C) = \alpha + \beta_1 \Delta_C^2 + \beta_2 \Delta_A^2 \Delta_C^2 + \beta_3 \Delta_B^2 \Delta_C^2 + \varepsilon,$$

Being $f(\Delta_A, \Delta_B, \Delta_C)$ the breakdown pressure response; Δ_A, Δ_B e Δ_C and the proportions of the types A, B and C the clays, respectively, and the error ε model.

Thus, the call function for this model and the ternary plot is:

```
%Triaxial(tab=gtens,Var=z,
x=A,y=B,w=C,
ctl=20 40 60 80 100,
col=white paoy lioy bio
deo,pxx=95,pxy=5,pyx=50,
pyy=97,pwx=5,pwy=5,help=1);
```



Figure 7. Grid call function macro and Ternary Graph for z1 response.

Through of the graph, you can see that the most parts of valid solutions show response to the resistance after burning is between 60 and 80 Pascal. Instead the best solution for this model was not pure clay type C shows that had low resistance to burning. However the best solutions are balanced mixture of C and A or mixture of clay and B, according withFigure 7. Furthermore the Triaxial Macro allows generate optional graphs to help understand the ternary graph.



Figure 8. Gradient Grid of first variable (A).



Figure 9. Gradient Grid of third variable (C).



Figure 10. Gradient Grid of second variable (B).



Figure 11. Grid of ternary graph: A,B and C variables.

CONCLUSION

A macro %Triaxial presents a general solution to ternary analysis using PROC GCONTOUR procedure to generate a tridimensional area. The macro transform the database informed and create a annotate database to display de grid lines of ternary graph. Auxiliary graphs are generated to help understand the ternary graph.

This paper suggests the importance of options into 3D procedures analysis to generate a nontraditional grid coordinate, as ternary graph. This problems is common into chemist, metallurgic and general industrial process and it would be interesting be found into SAS® procedures.

REFERENCES

Boldrino, J. L. 1980. *Álgebra Linear*. 3ª ed. São Paulo: Harper e Row do Brasil.

Campos, L. F. A., Amorim, L. V., Ferreira , H. C.**,** (2007), *Experimental design applied to the study of composition effect of bentonite on the rheology of drilling fluids. Part II: ternary compositions,* Federal University of Campina Grande, PB, Brazil.

Carpenter, A. L., (2005), *Storing and Using a List of Values in a Macro Variable*, Philadelphia, Pennsylvania: SUGI 30.

Linhares, Z. J., (2015) *Mass optimization of ceramics using design statistics with block of ceramic production*, Piauí Institute of Technology, PI, Brazil.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Igor Ferreira do Nascimento
Instituto Federal do Piauí - IFPI
+55 86 998084907
igor.nascimento@ifpi.edu.br

Roberto Arruda l. Soares
Instituto Federal do Piauí - IFPI
robertoarruda@ifpi.edu.br

Zacarias Linhares Junior
Instituto Federal do Piauí - IFPI
engenheirozacarias1@hotmail.com

APPENDIX I – SAS MACRO

```
/****************************************/
/********SAS MACRO TRIAXIAL*************/
/****************************************/
%macro wordcount(list);
        %* Count the number of words in
&LIST;
        %local count;
        %let count=0;

        %do %while(%qscan(&list, &count+1 ,
        %str( )) ne %str());
        %let count = %eval(&count+1);
        %end;
        &count
%mend wordcount;

%macro triangle(percx=100, pxx=95, pxy=5,
pyx=50, pyy=97, pwx=5, pwy=5,
                        help=1);
%let cos30=%SYSEVALF((3**(1/2))/2);
%let h=%SYSEVALF(&cos30.*&percx./100);
%let delta=3;%let by=5;
%let len=%SYSEVALF((100/&by.)*&percx./100);
%let bx=%SYSEVALF(50-10*(&percx./100));
%let byy=%SYSEVALF(&cos30.*&percx. + (100-
&cos30.*&percx)/2);

%if %sysfunc(exist(triangle)) %then
%do;
        proc delete data=triangle;
        run;
%end;
        %do i=1 %to &by.;
%do k=0 %to (&i-1);
data triangle_&i._&k.;
length function style color $ 8 text $ 25;
retain hsys xsys ysys "1";
when ="a";
/* create triaxial grid*/
function="move";
x=&bx. -((&i.-1)*(&len./2))+&k. * &len.;
y=&byy.-(((&i. - 0)*&h.)*&len.);
line=3;color="gray";size=.2;
transparency=0.2;variable=210;
output;
function="draw";
x=&bx. -((&i.-1)*(&len./2))+(&k.+1)* &len.;
y=&byy.-(&i.-0)*&h.*&len.;
transparency=0.2;        variable=13;
size=.2; line=3;
color="gray";output;
function="draw";
x=&bx. -((&i.-1)*(&len./2))+(&k.+1/2)*&len.;
y=&byy.-(&i.-1)*&h.*&len.;
size=.2;
variable=32;
output;
function="draw";line=3;
color="gray";
x=&bx. -((&i-1)*(&len./2))+&k.*&len.;
y=&byy.-(&i.-0)*&h.*&len.;
size=.2; variable=211;
output;run;

proc append base=triangle
data=triangle_&i._&k.;
proc delete data=triangle_&i._&k.;
run;

%end;%end;

data triangle_major;
length function style color $ 8 text $ 25;
retain hsys xsys ysys "1";when ="a";
function="move";
x=&bx. -((&by.-1)*(&len./2));
y=&byy.-(((&by. - 0)*&h.)*&len.);
line=1;color="gray";size=.2;
transparency=0.2;        variable=0;output;
function="draw"; x=&bx. -((&by.-
1)*(&len./2))+(&by.-1+1)* &len.;
y=&byy.-(&by.-0)*&h.*&len.;
transparency=0.2;variable=0;size=.2;
line=3;color="gray";output;
function="draw";
x=&bx. -((1-1)*(&len./2))+(0+1/2)*&len.;
y=&byy.-(1-1)*&h.*&len.;
size=.2; variable=0;output;
function="draw";line=3;
color="gray";
x=&bx. -((&by.-1)*(&len./2));
y=&byy.-(((&by. - 0)*&h.)*&len.);
variable=0;output;
run;

proc append base=triangle data=triangle_major;
proc delete data=triangle_&i._&k.;
run;

/* create annotate label of triaxial grid*/
data triangle_label;
length function style color $ 8 text $ 25;
retain hsys xsys ysys "1";when ="a";

%do i=1 %to %SYSEVALF(&by-1);
function="label";
x=&bx. -((&i.-1)*(&len./2))-&delta.;
y=&byy.-(&i.-0)*&h.*&len.;
text=LEFT(%SYSEVALF((100/&by.)*(&i.)) || "%");
transparency=0.5;
size=2;
variable=3;
color="black";
output;
function="label";
x=&bx. +((&i.+1)*(&len./2)) + &delta.;
y=&byy.-(&i.-0)*&h.*&len.;
text=LEFT(%SYSEVALF((100/&by.)*(5-&i.)) ||
"%");
transparency=0.5;size=2;
variable=1;color="black";
output;
function="label";x=&bx. +(&i.-2)*&len.;
y=&byy.-&by.*&h.*&len.-&delta./2;
text=LEFT(%SYSEVALF((100/&by.)*&i.) || "%");
transparency=0.5;size=2;variable=2;
color="black";output;
%end;
        function="label";
        x=&pxx.; y=&pxy.;
        text="&x.";variable=2;
        transparency=0.5;
        size=4;color="black";
        output;function="label";
        x=&pwx.; y=&pwy.;
        variable=3;text="&w.";
        transparency=0.5;
        size=4;color="black";
        output;function="label";
        x=&pyx.; y=&pyy.;
        text="&y.";variable=1;
        transparency=0.5;
        size=4;color="black";
        output;
run;

proc append base=triangle data=triangle_label;
proc delete data=triangle_label;
run;
run;
quit;
%if(&help.=1) %then
        %do;
*Ploting first variable grid;
data trianglex;set triangle;
        where (variable=2 or variable=32 or
variable=211 or variable=0);
if variable=211 then
function="draw";
if variable=32 then function="move";
run;

*Ploting second variable grid;
data triangley;  set triangle;
where (variable=1 or variable=210 or
variable=13 or variable=0);
```

```
if variable=210 then function="move";
run;

*Ploting third variable grid;
data trianglew;  set triangle;
where (variable=3 or variable=13 or
variable=32 or variable=0);
if variable=13 then function="move";
if variable=32 then function="draw";
run;

proc ganno annotate=trianglex;
proc ganno annotate=triangley;
proc ganno annotate=trianglew;
proc ganno annotate=triangle;
run;
%end;
%mend triangle;
%macro Triaxial(tab=, /*the name of database*/
Var=, /*the name of the variable to be
analyzed*/
x=, /*the name of the variable in first axis
of ternary graph*/
y=, /*the name of the variable to be used to
generate the second axis of ternary graph*/
w=, /*the name of the variable to be used to
generate the third axis of ternary graph*/
ctl=, /*The vector with the cut lines of
contour analysis*/
col=, /*The vector with the colors of cut
lines of contour analysis*/
pxx=95, /*Label position of first variable in
axis x in annotate coordinate system 1, 0-
100*/
pxy=5, /*Label position of first variable in
axis y in annotate coordinate system 1, 0-
100*/
pyx=50, /*Label position of second variable in
axis x in annotate coordinate system 1, 0-
100*/
pyy=97, /*Label position of second variable in
axis y in annotate coordinate system 1, 0-
100*/
pwx=5, /*Label position of third variable in
axis x in annotate coordinate system 1, 0-
100*/
pwy=5, /*Label position of third variable in
axis y in annotate coordinate system 1, 0-
100*/
help=1);/*Dichotomy variable to generate
additional graphs to help understand the
ternary graph. 0: not and 1 yes, 1 default*/
%let cos30=%SYSEVALF((3**(1/2))/2);
%let percx=100;
%let limx1=%SYSFUNC(ROUND(0-(100-&percx.)/200,
.001));
%let limx2=%SYSFUNC(ROUND(1+((100-
&percx.)/200), 0.001));
%let limy1=%SYSFUNC(ROUND(0-(1-
&cos30.)*(&limx2.-&limx1.)/2, 0.001));
%let limy2=%SYSFUNC(ROUND(&cos30.*&percx./100+
(1-&cos30.)*(&limx2.-&limx1.)/2,0.001));

proc g3grid data=&tab. out=g&tab.;
grid &x.*&y.=&var./ axis1=-1 to 2 by .01
axis2=-1 to 2 by .01;
run;quit;
PROC SQL NOPRINT;
SELECT MAX(&VAR.)into:max FROM g&tab.;
SELECT MIN(&VAR.)into:min FROM g&tab.;
RUN;QUIT;
DATA g&tab.;SET g&tab.;
A1=&x.  + &y.*0.5;
B1=&y. * &cos30.;
IF ((A1 - &y.*0.5)<0) THEN &var.=&min.;
IF ((A1 + &y.*0.5)>1) THEN &var.=&min.;
IF &y.<0 THEN &var.=&min.;
RUN;
proc g3grid data=g&tab. out=pg&tab.;
grid A1*B1=&var./ axis1=&limx1. to &limx2. by
.001
axis2=&limy1. to &limy2. by .001;
run;
quit;
PROC SQL NOPRINT;
SELECT MAX(&VAR.)into:max FROM pg&tab.;
SELECT MIN(&VAR.)into:min FROM pg&tab.;
RUN;QUIT;
data pg&tab.;set pg&tab.;
if b1<0 or b1>&cos30. Then &var.=&min.;
if a1<0 or a1>1 then &var.=&min.;
run;
%triangle(percx=&percx., pxx=&pxx., pxy=&pxy.,
pyx=&pyx., pyy=&pyy.,
pwx=&pwx., pwy=&pwy., help=&help.);
axis1 LABEL=NONE minor=none value=none
label=none;
goptions /*reset=global*/
gunit=pct border cback=white transparency
colors=(&col.);
proc gcontour data=pg&tab. ANNOTATE=triangle;
plot B1*A1=&var./ haxis=axis1 vaxis=axis1
levels=&ctl. pattern join;
run;quit;

%mend Triaxial;
```